Introduction to Dynamical Systems

A HANDS-ON APPROACH WITH MAXIMA



Jaime E. Villate

ISBN 972-99396-0-8

9 789729 939600

Introduction to Dynamical Systems

A Hands-on Approach with Maxima

Jaime E. Villate
College of Engineering
University of Porto, Portugal

Introduction to dynamical systems: a hands-on approach with Maxima

E-mail: villate@fe.up.pt

ISBN: 972-99396-0-8

This is a partial translation of Portuguese version 1.1 of November 23, 2006

The cover figure is the Julia set for the complex number $-0.75 + i\,0.1$, with 48 iterations, as explained in chapter 12.

Contents

List of Figures

Preface

This book is updated very often. The number of the current version can be found on the second page and the most recent version can always be found on the Web at http://fisica.fe.up.pt/maxima/dynamicalsystems. This version has been written to be used with Maxima's version 5.10. (http://maxima.sourceforge.net).

This book started as the lecture notes for a one-semester course on the physics of dynamical systems, taught in the period 2003-2006 at the College of Engineering of the University of Porto. The field of dynamical systems is at the borderline of physics, mathematics and computing. Being a scientific field that is not taught with the traditional axiomatic method used in other physics and mathematics courses, it is appropriate to use a practical teaching approach based on projects done with a computer.

The study of dynamical systems advanced very quickly in the decades of 1960 and 1970, giving rise to a whole new area of research with an innovative methodology that gave rise to heated debates within the scientific community. The innovative impulse was ignited by the rapid development of computers.

A new generation of researchers was born, who used their computers as laboratories for experimenting with equations discovering new phenomena. The traditional mathematics criticized that approach for the lack of a rigorous mathematical theory that explained those new results. Many of those results were found within the framework of physical problems: non-linear dynamics, condensed-matter physics and electromagnetism. However, many physicists regarded that new research field simply as a computer simulation of old physical concepts long established, without any new physics in it. A usual comment would be: "this is all very nice, but where is the physics?".

Thus, the pioneers in this new field of dynamical systems would be often confronted with rejected publications in scientific journals and negative assessments of their work. On the other hand, their activities awakened a strong interest that grew up exponentially and was viewed by some as a fresh air current; the methods used in the study of dynamical systems match well with the working environment of modern scientists.

The new paradigm spread to teaching and the traditional courses on physics and mathematics have been gradually "infected" with this new experimental/ computational methodology, contrasting with the traditional axiomatic method. As it happened in the scientific community, the new methodology has also led to some debate among teachers; at the same time, it has awakened big interest as a better way to motivate today's students. Subjects such as chaos and fractals are very appealing to students.

In this book we intend to explore some topics on dynamical systems, using an active teaching approach, supported by computing tools and trying to avoid too may abstract details. The use of a

Computer Algebra System (CAS) does not eliminate the need for mathematical analysis from the student; using a CAS to teach an engineering course does not turn it into a purely technical subject either. One of the difficulties inherent to any Computer Algebra System is the fact that there are no unique solutions to the problems they solve. Different methods to solve a problem may lead to solutions that look very different but might be equivalent. Or the solutions can be really different that are only equivalent within some domain. In some cases the system does not give any answer or it might even give a wrong answer.

It is necessary to gain some experience to be able to use CAS tools successfully and to be able to test the validity of its results. In the process of gaining that experience, the user will also gain a better insight into the mathematical methods used bay the system.

Nowadays the great majority of engineering and exact sciences professionals depend on a calculator to calculate the square root of a real number, for instance, 3456. Some of us were taught in School how to do that with pencil an paper, in a time when there were no calculators. I do not believe that this new dependence is a serious handicap, and I'm not in favor of teaching kids how to calculate square roots with paper and pencil before they are allowed to use the calculator. What I find very important is that the algorithm we used to calculate square roots with paper and pencil remains available and well documented in the literature; it is an important piece in our legacy of algorithms.

On the other hand, now that students have calculators to compute square roots, they can move faster into other topics such as the study of quadratic equations; and in doing so, they might even gain a deeper insight of the function $\sqrt{x}$, which they did not attain when they had to spend a lot of time learning the algorithm to calculate square roots. In the case of differential equations and difference equations, with the help of Computer Algebra Systems students can advance faster into subjects such as chaos and fractals, instead of dedicating a whole semester to learn several algorithms to obtain analytical solutions for a few types of equations.

I would like to acknowledge the help of my colleagues Helena Braga and Francisco Salzedas, with whom I have taught the course on Physics of Dynamical Systems; I would also like to thank our students in that course throughout the last 3 years; their positive comments have encouraged me to undertake the task of writing this book. The students have been asked to make projects for that course, and some of those projects were very interesting and helped me learn some of the subjects covered in this book. Special thanks go to the student Pedro Martins who made a careful review of the manuscript.

Jaime E. Villate
Porto, November 2006

Chapter 1

Introduction

1.1 Differential equations

Differential equations play a very important role in Engineering and Science. Many problems lead to one or several differential equations that must be solved. Most attention has been given to linear equations in the literature; several analytical methods have been developed to solve that type of equations.

Non-linear differential equations are much harder to analyze and there are no general solution techniques for those equations. Problems that lead to linear equations are easier to study. From the last half of the 20th century, the rapid development of the computer made it possible to solve non-linear problems using numerical methods. Non-linear systems lead to a wealth of new and interesting phenomena not present in linear systems.

A new approach, that relies more on geometric interpretation rather than analytical analysis, has gained popularity for the study of non-linear systems. Many of the concepts used in that geometrical approach, such as the phase space, have long be used in dynamics to study the motion of a mechanical system.

In order to give a short introduction to that methodology to study differential equations, in the next chapters we will consider several problems specific dynamics and electrical circuit theory. Before we begin, we will introduce a Computer Algebra System (CAS), Maxima, which will be used extensively throughout the book.

1.2 Solving physics problems with Maxima

Maxima is a software package in the category of Computer Algebra Systems (CAS), namely, a system that can be used not just for numerical calculation but also to deal with algebraic equations with abstract variables. There are various CAS packages available; we have decided to use Maxima because it is Free Software. That means that it can be installed and used by our students without having to obtain a license for it, and they can even study its source code to get a better understanding of how that system works. Another important advantage is the possibility of modifying the original package to better suit our needs; we took advantage of that facility to add new

features needed for this book.

Maxima includes several functions to manipulate mathematical functions, including differentiation, integration, power series approximation, Laplace transforms, solving ordinary differential equations and graph plotting in 2 and 3 dimensions. It can also work with matrices and vectors. Maxima can be used to solve problems numerically and write down programs as done with traditional programming languages.

The following examples should serve to give an first glimpse at the way Maxima can be used. In the next chapters we will go deeper into the subject, but readers who are not familiar with Maxima and would like to have a general overview from the beginning can start by going through appendix A. The examples that we will solve in this section are in the area of dynamics of a particle and direct-current circuits, which are the main subjects in this book. A minimum knowledge in those two subjects will be necessary in order to follow those examples.

Example 1.1

A battery is connected to an external resistor with resistance R and the voltage across the resistor is measured with a voltmeter V. To find the electro-motive force ε *and the internal resistance r of the battery, two external resistors of 1.13 k*Ω *and 17.4 k*Ω *were used. The voltage drop in both cases were 6.26 V e 6.28 V. Find the intensity of the current in both cases. Obtain the values of* ε *and r. Plot a graph of the power dissipated in the external resistance, as a function of R, for values of R within 0 and* $5r$.



Solution: The current through R is found from **Ohm's law**:

$$I = \frac{\Delta V}{R} \tag{1.1}$$

With the values given for the potential difference, ΔV, and the resistance, R, we can use Maxima to find the currents:

```
(%i1) 6.26/1.13e3;
(%o1)                          .005539823008849558
```

When Maxima's console is started, the `(%i1)` label appears indicating that the system is ready to accept a command; the letter "i" stands for *input*. The expression 1.13e3 is the form used to represent the number 1.13×10^3 in Maxima. Each command must end with a semi-colon. When the "Enter" key is pressed, the system responds with a label `(%o1)` followed by the result of the first command `(%i1)`; "o" stands for *output*.

The current in the second case is computed in a similar way:

```
(%i2) 6.28/17.4e3;
(%o2)                          3.609195402298851E-4
```

Thus, the current in the 1.13 kΩ resistor is 5.54 mA, and in the 17.4 kΩ resistor is 0.361 mA.

To obtain the battery's electro-motive force and internal resistance we should use the **voltage-current characteristic** for a battery:

$$\Delta V = \varepsilon - rI \tag{1.2}$$

replacing the two set of values given for ΔV and R we will get a system of two equations with two variables. We will save those two equations in two Maxima variables that we will dub as `eq1` and `eq2`

```
(%i3) eq1: 6.26 = emf - r*%o1;
(%o3)                  6.26 = emf - .005539823008849558 r
(%i4) eq2: 6.28 = emf - r*%o2;
(%o4)                  6.28 = emf - 3.609195402298851E-4 r
```

notice that the symbol used to save a value in a variable is a colon and not an equal sign. A maxima variable can store a numerical value or something more abstract as a mathematical equation in this case. The equal sign makes part of the equation that is being stored. To avoid having to write the numerical values of the currents obtained previously, we used the symbols `%o1` and `%o2` that stand for the value of those previous results.

The last two equations constitute a linear system of equations with two variables. That kind of system can be solved in Maxima, using the command `solve`:

```
(%i5) solve([eq1,eq2]);
                                 983100        79952407
(%o5)                    [[r = ------, fem = --------]]
                                 254569        12728450
(%i6) %,numer;
(%o6)          [[r = 3.861821352953423, fem = 6.281393806787158]]
```

The syntax `[eq1,eq2]` was used to create a **list** with two elements, which is what the command `solve` expects when there are more than one equation to be solved. Some warning messages given by Maxima were omitted above. The command `solve` gives an exact result, in the form of two rational numbers. The command in `%i6` was used to approximate those rational numbers with fixed-point numbers. The symbol `%` stands for the output of the last command executed; in this case it is equivalent to `%o5`. We thus conclude that the electro-motive force is approximately 6.2814 V and the internal resistance is 3.8618 Ω.

The electric power dissipated in the resistance R is

$$P = RI^2$$

the current I across the external resistor can be calculated in terms of the electro-motive force and the resistances r and R

$$I = \frac{\varepsilon}{R+r}$$

therefore, the power dissipated in the external resistor is

$$P = R\left(\frac{\varepsilon}{R+r}\right)^2$$

to draw the plot of P, as a function of R, we can use the following command

```
(%i7) plot2d(R*(6.2814/(R+3.8618))^2, [R, 0, 5*3.8618]);
```

the result is shown in figure 1.1. Moving the cursor in the graphic window, it is possible to read the coordinates of the point where the cursor is. We can check that the power dissipated in external resistor takes its maximum value when the external resistance equals the internal one.



Figure 1.1: Power dissipated in the external resistor as a function of the external resistance.

Two frequent errors should be avoided:

- An expression such as

  ```
  a = 3;
  ```

  will not assign any value to the variable a. After that expression is written, variable `a` remains as an undefined variable. To assign a value of 3 to the variable `a`, the following command must be used:

  ```
  a : 3;
  ```

- New users of Maxima should become aware with the distinction between equations and expressions. An example of an equation is

```
x^2 - 3*x = 2*x + 5
```

while an expression is something such as

```
2*x + 5
```

Some commands in Maxima accept only equations or expressions as their input values. For instance, the command `plot2d` used in the previous example accepts only expressions and not equations. The command `solve` requires one equation, or a list with several equations, but it will also accept expressions instead of equations: each expression given will be automatically converted into an equation by equating it to zero; for instance, the command

```
solve(x^2 - 5*x + 5);
```

will find the two values of x that will solve the equation

$$x^2 - 5 * x + 5 = 0$$

Example 1.2
The position vector for a particle, as a function of time t, is given by the equation:

$$\vec{r} = \left(5 - t^2\, e^{-t/5}\right) \vec{e}_x + \left(3 - e^{-t/12}\right) \vec{e}_y$$

in MKS units. Find the vectors for the position, velocity and acceleration in the instants $t = 0$, $t = 15$ s, and when time approaches infinity. Plot the trajectory of the particle during the first 60 seconds of its motion.

Solution: We start by representing the position vector as a list with two elements; the first element will be the x coordinate and the second one will be the y coordinate. We will save that list in a variable named r, so we can use it later on.

```
(%i8) r: [5-t^2*exp(-t/5),3-exp(-t/12)];
                              2   - t/5        - t/12
(%o8)                   [5 - t  %e     , 3 - %e      ]
```

the vector velocity equals the derivative of the position vector and the vector acceleration is the derivative of the vector velocity. The command used in Maxima to find the derivative of an expression is `diff`. The input to that command can also be a list with several expressions; thus, the velocity and acceleration are:

```
(%i9) v: diff(r,t);
                       2   - t/5                  - t/12
                      t  %e              - t/5  %e
(%o9)                [---------- - 2 t %e     , --------]
                          5                        12
(%i10) a: diff(v,t);
```

```
                  2   - t/5         - t/5                   - t/12
                 t  %e        4 t %e            - t/5    %e
(%o10)       [- ---------- + ----------- - 2 %e     , - --------]
                    25           5                        144
```

the constant `%e` in Maxima represents the Euler number, e. To find the position, velocity and acceleration in the instant $t = 0$, we use the following commands

```
(%i11) r, t=0, numer;
(%o11)                              [5, 2]
(%i12) v, t=0, numer;
(%o12)                      [0, .08333333333333333]
(%i13) a, t=0, numer;
(%o13)                    [- 2, - .006944444444444444]
```

The argument `numer`, was used to get the result in floating-point form. In vector notation, the results we obtained are:

$$\begin{aligned}\vec{r}(0) &= 5\vec{e}x+2\vec{e}_y \\ \vec{v}(0) &= 0.08333\,\vec{e}_y \\ \vec{a}(0) &= -2\vec{e}x-0.006944\,\vec{e}_y\end{aligned}$$

For $t = 15$ s the results are obtained in a similar way

```
(%i14) r, t=15, numer;
(%o14)              [- 6.202090382769388, 2.71349520313981]
(%i15) v, t=15, numer;
(%o15)              [.7468060255179592, .02387539973834917]
(%i16) a, t=15, numer;
(%o16)                     [0.0497870683678639, - .001989616644862431]
```

The limiting values when times goes to infinity can be calculated with Maxima's command `limit`; the symbol used in Maxima to represent infinity is `inf`

```
(%i17) limit(r,t,inf);
(%o17)                              [5, 3]
(%i18) limit(v,t,inf);
(%o18)                              [0, 0]
(%i19) limit(a,t,inf);
(%o19)                              [0, 0]
```

Thus, a particle will approach the point $5\vec{e}_x + 3\vec{e}_y$, where it will remain at rest.

To plot the graph of the trajectory we will use the option `parametric` of the command `plot2d`. the x and y components of the position vector will be given separately; the command `plot2d` will not accept them inside a list as we have been using them. To get the first element of the list `r` (x component) is labelled as `r[1]` and the second element `r[2]`.

```
(%i20) plot2d([parametric,r[1],r[2],[t,0,60],[nticks,100]]);
```

The time domain, from 0 to 60, is defined with the notation `[t,0,60]`. The option `nticks` was used to increase the number of intervals of t, because the default value of 10 intervals would render a broken curve instead of a continuous trajectory. The graph obtained is shown in figure 1.2.



Figure 1.2: Trajectory of the particle during the first 60 seconds, from the initial instant when t was equal to 5.2 .

1.3 References

For more information about Maxima, see appendix A and the Maxima Book (de Souza et al., 2003).

1.4 Multiple-choice questions

1. Only one of the following Maxima commands is correct. Which one?

A. `solve(t-6=0,u-2=0,[t,u]);`

B. `solve(t+4=0,u-4=0,t,u);`

C. `solve([x^3+4=2,y-4],x,y);`

D. `solve(x-6=0,y-2=0,[x,y]);`

E. `solve([t+3,u-4],[t,u]);`

2. Newton's second law was defined in Maxima with:

```
(%i6) F = ma;
```

which Maxima command should be used to compute the value of the force corresponding to a mass of 7 with an acceleration of 5 (SI units).

A. `solve(F, m=7, a=5);`

B. `solve(F, [m=7, a=5]);`
C. `solve(%o6, m=7, a=5);`
D. `%o6, m=7, a=5;`
E. `solve(F: m=7, a=5)`

3. If we input the following commands in Maxima:

```
(%i1) x:3$
(%i2) x=5$
(%i3) x;
```

which will be the output (%o3)?

A. 5
B. x
C. 3
D. true
E. 0

1.5 Problems

1. An amperimeter was use to measure the current at points D and F in the circuit whose diagram is shown in the figure. At point D the value obtained for the current was 0.944 mA, in the direction ADC, and at point F it was 0.438 mA, in the direction CFE. (*a*) Store the equation for Ohm's law, $V = IR$, in a Maxima's variable "ohm". (*b*) Give a value to variable I equal to the current at point D, and substitute the resistance in Ohm's law with each of the values 2.2 kΩ and 6.8 kΩ, to compute the potential difference in each of the resistors; repeat the same procedure to calculate the potential difference in each resistor.



2. The position of a particle moving along the x axis is given by the equation $x = 2.5t^3 - 62t^2 + 10.3t$, where x is measured in meters and t in seconds. (*a*) Find the expressions for the velocity and the acceleration as a function of time. (*b*) Find the values of the time, position and acceleration in all the instants when the particle is at rest ($v = 0$). (*c*) Draw the plots for the position, velocity and acceleration as a function of time, for t in the interval between 0 and 20 seconds.

3. The position vector of a particle, as a function of time, is given by the equation:

$$\vec{r} = \left(5.76 - \mathrm{e}^{-t/2.51}\right)\vec{e}_x + \mathrm{e}^{-t/2.51}\cos(3.4t)\,\vec{e}_y$$

in SI units. (*a*) Compute the position, velocity and acceleration in the instants t=0, t=8 s, and when time goes to infinity. (*b*) Plot the graphs of the x and y components of the position, as a function of time, for t in the interval between 0 and 15 seconds. (*c*) Plot the graph of the trajectory, on the plane xy, in the interval of t between 0 and 15 s.

Chapter 2

Discrete dynamical systems

A discrete dynamical system is a system whose state only changes during a discrete sequence of instants $\{t_0, t_1, t_2, \ldots\}$. In the interval between two of those instants the state remains constant.

In this chapter we will consider only discrete systems in one dimension as an introduction to the study of continuous dynamical systems in the next chapters. We will come back to the subject of second-order discrete systems in a later chapter. The state of a one-dimensional discrete system is defined by a variable y. The values of that variable during the instants $\{t_0, t_1, t_2, \ldots\}$ will be a sequence $\{y_0, y_1, y_2, \ldots\}$ (that sequence is dubbed an **orbit** of the system). The interval of time between different pairs of consecutive instants t_n and t_{n+1} does not have to be the same.

The **evolution equation** allows us to compute the state y_{n+1}, at an instant t_{n+1}, from the value of the state y_n at the previous instant t_n:

$$y_{n+1} = F(y_n) \tag{2.1}$$

where $F(y)$ is some known function. The equation above is a **difference equation** of the first order. Given an initial state y_0, successive application of the function F will generate the sequence of states y_n which determine the evolution of the system. In some cases it might be possible to find a general expression for y_n as a function of n.

Example 2.1
Fond the first four terms in the evolution of the system $x_{n+1} = \cos x_n$, with initial state $x_0 = 2$

Solution: Applying the difference equation three times, we obtain the first four terms in the sequence:

$$\{2,\ \cos(2),\ \cos(\cos(2)),\ \cos(\cos(\cos(2)))\} \tag{2.2}$$

Example 2.2
A loan of $ 500 is obtained from a bank, which charges a 5% yearly interest rate. The loan is to be paid in 20 months, with monthly payments of $ 26.11. What will be the amount in debt after 10 months?

Solution: During the month n the amount in debt, y, will be equal to the amount in debt in the previous month, y_{n-1}, plus the interests due for that month, minus the monthly payment p:

$$y_n = y_{n-1} + j\,y_{n-1} - p \tag{2.3}$$

where j is the monthly interest rate (0.05/12). Using Maxima, the sequence of amounts in debt y_n can be obtained by applying the above recurrence relation several times:

```
(%i1) j: 0.05/12$

(%i2) y: 500$

(%i3) y: y + j*y - 26.11;

(%o3)                          475.9733333333333
(%i4) y: y + j*y - 26.11;

(%o4)                          451.8465555555555
(%i5) y: y + j*y - 26.11;

(%o5)                          427.619249537037
```

it would be necessary to repeat the command `(%i3)` ten times. Another method consists in defining a function depending on an integer variable, using the recurrence relation, and to use that function to calculate y_{10} directly:

```
(%i6) y[0]: 500$

(%i7) y[n] := y[n-1] + j*y[n-1] - 26.11;

(%o7)                  y  := y      + j y      - 26.11
                        n     n - 1      n - 1
(%i8) y[10];

(%o8)                          255.1779109580579
```

Some care should be taken in Maxima when using functions of a integer argument. In the previous example, when we calculated `y[10]`, the values of `y[9]`, `y[8]`,..., `y[1]`, were also implicitly calculated and stored in memory. If we changed the recurrence relation, those values that were already calculated and stored would not be updated. Thus, before we change the recurrence relation, or the initial value `y[0]`, it is necessary to erase the previously calculated sequence, by using the command `kill`.

For example, if the value of the loan was duplicated to $ 1000, and the monthly payment was also duplicated, will the amount in debt after the tenth payment would also duplicate? let us see:

```
(%i9) kill(y)$
```

```
(%i10) y[0]: 1000$

(%i11) y[n] := y[n-1] + j*y[n-1] - 52.22;

(%o11)                  y  := y      + j y      - 52.22
                         n     n - 1      n - 1
(%i12) y[10];

(%o12)                          510.3558219161157
```

thus, the amount in debt also doubles.

Another question that we might ask in the original example is: what will the monthly payment should be if instead of 20 months we would want to pay the loan in 40 months?

To answer that question, we use a variable p to represent the monthly payment, we calculate the amount in debt after 40 months, as a function of p, and we equal that expression to zero to calculate the value of p.

```
(%i13) kill(y)$

(%i14) y[0]: 500$

(%i15) y[n] := expand(y[n-1] + j*y[n-1] - p)$

(%i16) solve(y[40] = 0, p);

RAT replaced 590.4757124853373 by 352514//597 = 590.4757118927973

RAT replaced -43.428341992962 by -26969//621 = -43.4283413848631

                                      72970398
(%o16)                          [p = --------]
                                      5366831
(%i17) %, numer;

(%o17)                      [p = 13.59655222979818]
```

The monthly payment should be $ 13.60. The function `expand` had to be used to force Maxima to calculate the products in every step, avoiding large expressions with several parenthesis in the calculation of y_n. The additional messages written by Maxima tell us that some floating point numbers were represented as fractions, to prevent numerical errors.

2.1 Discrete systems evolution

The evolution of a first-order discrete system:

$$y_{n+1} = F(y_n) \tag{2.4}$$

is obtained by applying successively a function F, to the initial state $y_0 = c$:

$$\{c, F(c), F(F(c)), F(F(F(c))), \ldots\} \tag{2.5}$$

or in a more compact form:

$$\{c, F(c), F^2(c), F^3(c), \ldots y_n = F^n(c)\} \tag{2.6}$$

2.2 Graphical analysis

A graphical method to represent the evolution of a system consists on plotting a point for each step in the sequence, with x-coordinate equal to the index n and y-coordinate equal to y_n. In Maxima, the function `evolution` in the additional package `dynamics` will plot such a graph.

Three arguments should be given to that program. The first argument must be an expression that depends only on the variable y; that expression will specify $F(y)$ from the right hand side of the difference equation 2.1. The second argument should be the initial value y_0, and the third argument is the number of elements, of the sequence, that should be plotted.

For instance, in example 2.1, using the variable y, we have $F(y) = \cos y$, with initial value $y_0 = 2$. To obtain the evolution graph with the first 20 terms, we use the commands:

```
(%i18) load("dynamics")$

(%i19) evolution(cos(y), 2, 20)$
```

The graph obtaine i `(%i19)` is shown in figure 2.1.



Figure 2.1: Evolution of $y_{n+1} = \cos(y_n)$ with $y_0 = 2$.

Another type of diagram which will be very useful to analyze discrete dynamical systems in one dimension is the so-called **staircase diagram**,[1] which consists in plotting the functions $y = F(x)$ and $y = x$, as well as an alternating sequence of horizontal and vertical segments joining the points



Figure 2.2: Staircase diagram for $x_{n+1} = \cos(x_n)$ with $x_0 = 2$.

(y_0,y_0), (y_0,y_1), (y_1, y_1), (y_1,y_2), and so on. For example, figure 2.2 shows the staircase diagram for the sequence represented in figure 2.1

The function `staircase`, included in the additional package `dynamics`, plots staircase diagrams. That function needs the same three arguments as the function `evolution`; namely, function $F(y)$ from the right-hand side of the difference equation 2.1, the initial value y_0 and the number of steps in the sequence. Notice that the independent variable in the expression for F should always be y. You might need to make the appropriate change if the state variable is something different from y in your problem.

For example, the graph 2.2 was obtained with the command

```
(%i20) staircase(cos(y),2,8)$
```

Notice that we did not have to load the package `dynamics` again, because it was already loaded in `(%i18)`. the staircase diagram allows us to understand when a sequence will converge or diverge. For instance, consider the system $y_{n+1} = y_n^2 - 0.2$. If we start with a value $y_0 = 1.1$, we obtain the graph on the left side of figure 2.3; we see that the sequence will converge to a negative value y, which is at the intersection of the functions $F(y) = y^2 - 0.2$ and $G(y) = y$, namely, $y = (5-3\sqrt{5})/10$.

The two functions intersect in another point, positive, equal to $y = (5-3\sqrt{5})/10$. In the graph we can see that even though the initial value was close to the second intersection point, the sequence moved away from it and towards the second intersection point, due to the fact that between the two intersection points the function $y^2 - 0.2$ is under $G(y) = y$. if we used an initial value to the right of the second intersection point, for instance, $y_0 = 1.5$, the sequence grows quickly towards infinity (right-hand side in figure 2.3). To make the sequences converge to the second intersection point, it would have been necessary that between the two intersection points $F(x)$ were above $G(y) = y$; that is to say, the slope of $F(y)$ should be less than 1, rather that greater than 1, at the second intersection point.

[1] Also know as cobweb diagram.



Figure 2.3: Solution to the system $y_{n+1} = y_n^2 - 0.2$ with initial values 1.1 (left) and 1.5 (right).

Example 2.3
Analyze the solutions to the logistic model, which consists in considering a population P with constant natality rate, a, and a mortality rate directly proportional to the population, bP, where a and b are constants.

Solution: The population under study could be a group of specimens from some animal species, where the sequence $\{P_0, P_1, P_2, \ldots\}$ represents represents the number of specimens throughout several consecutive years.

Let P_n represent the number of specimens at the beginning of period n. during that period of time, an average aP_n new specimens are born, and bP_n^2 specimens die. Thus, in the beginning of the next period, $n+1$, the population will be

$$P_{n+1} = (a+1)P_n\left(1 - \frac{b}{a+1}P_n\right) \tag{2.7}$$

It is convenient to define a variable y such as $y_n = bP_n/(a+1)$. Thus, we obtain an equation with a single parameter $c = a+1$

$$y_{n+1} = cy_n(1-y_n) \tag{2.8}$$

Figure 2.4 shows the solutions obtained with an initial value $y_0 = 0.1$, in the cases $c = 2$ and $c = 4$. With $c = 2$, the solution converges quickly to the fixed point $y = 0.5$.

With $c = 4$, the state of the system goes through many different values between 0 and 1, and it does not seem to follow any regular pattern. That type of behavior is called **chaotic**. the state in any given period is perfectly determined from the state in the previous period, but a very small modification of the state in an initial period will lead to a completely different pattern in later periods.

2.3 Stationary points

A **stationary point** of the system 2.1 is a point c at which the state of the system would stay constant. For that to happen, it will necessary and sufficient that

$$F(c) = c \tag{2.9}$$



Figure 2.4: Solutions of the logistic model with an initial value 0.1. With $c = 2$ (left) the sequence converges, while with $c = 4$ (right) it becomes chaotic.

From the graphical point of view, the stationary points are all those points where the curve $F(x)$ intersects the right line $y = x$ in the stairway diagram. For example, in the case of he logistic model, figure 2.4 shows that in the two cases $c = 2$ and $c = 4$ there are two stationary points, one of them at $y = 0$. We can use Maxima's command `solve` to find the stationary points; in the case $c = 4$, it goes like this

```
(%i21) flogistic: 4*y*(1-y);

(%o21)                              4 (1 - y) y
(%i22) stationary: solve(flogistic - y);

                                     3
(%o22)                         [y = -, y = 0]
                                     4
```

The two stationary points are 0 and 0.75.

Let us consider a stationary point, where the curve $y = F(x)$ intersects the straight line $y = x$, such that the derivative $F'(y)$ is bigger than 1 at that point. Namely, as the curve $F(x)$ crosses the straight line from left to right, it goes from under the line to above it, and if we draw the staircase diagram starting at a point near that stationary point, the sequence will move away from that point. We say that the stationary point is repulsive, or unstable.

Index

Bibliography

de Souza, P. N., Fateman, R., Moses, J. & Yapp, C. (2003), *The Maxima book*, http://maxima.sourceforge.net.